

TA: Jade Cheng
ICS 312
Homework Solution #14
Due Date: October 08, 2009

File Processing Homework Code

```
;; Program construct a text file which contains:
;;     Humpty Dumpty sat on the wall.
;;     Humpty Dumpty had a great fall.
;; Then, program uses the function 3fh to read this file into a buffer, and then
;; use function 40h to write the buffer out to the screen.
        include pcmac.inc
        .model      small
        .stack 100
        .data

fname          db      'myfile.txt', 0
handle         dw      ?
msg            db      'Humpty Dumpty sat on the wall.', 0dh, 0ah,
                    'Humpty Dumpty had a great fall.'

buffer         db      100 dup (?)
create_error_found db 0dh, 0ah, 'create error found.$'
write_error_found db 0dh, 0ah, 'write error found.$'
read_error_found  db 0dh, 0ah, 'read error found.$'
close_error_found db 0dh, 0ah, 'close error found.$'
seek_error_found db 0dh, 0ah, 'seek error found.$'

        .code
prog  proc
      extrn      putdec : near    ; compile it with util.lib
      extrn      getdec : near
      mov        ax, @data
      mov        ds, ax

      mov        ah, 3ch          ; create file function
      lea        dx, fname        ; copy address to dx
      mov        cl, 0             ; normal file attributes
      int        21h              ; create the file
      jc         create_error     ; jump if error
```

```

    mov     handle, ax      ; store file handle

    mov     ah, 40h        ; write file
    mov     bx, handle     ; file handle for the opened file
    mov     cx, 63         ; length of message
    lea     dx, msg        ; copy offset to dx
    int     21h           ; write the msg to the file
    jc     write_error    ; jump if error

    mov     ah, 42h        ; move file pointer
; mov     bx, handle     ; (still the file handle)
    mov     cx, 0          ; clear cx
    mov     dx, 0          ; 0 bytes to move
    mov     al, 0          ; relative to the beginning
    int     21h           ; move the file pointer
    jc     seek_error     ; jump if error

    mov     ah, 3fh        ; read the file
; mov     bx, handle     ; (still the file handle)
    mov     cx, 63         ; set count to read
    lea     dx, buffer     ; move offset of buffer to dx
    int     21h           ; read to buffer, ax = bytes read
    jc     read_error     ; jump if error

    mov     ah, 3eh        ; close file function
; mov     bx, handle     ; (still the file handle)
    int     21h           ; close the file
    jc     close_error    ; jump if error

    mov     ah, 40h        ; write to screen
    mov     bx, 1          ; file handle for the stdout
    mov     cx, 63         ; length of message
    lea     dx, buffer     ; copy offset to dx
    int     21h           ; display msg
    jc     write_error    ; jump if error
    _putch 0ah            ; prints a blank line
    jmp     done

create_error:                ; error handles
    _putstr create_error_found
    jmp     done
write_error:

```

```
        _putstr    write_error_found
        jmp      done
read_error:
        _putstr    read_error_found
        jmp      done
close_error:
        _putstr    close_error_found
        jmp      done
seek_error:
        _putstr    seek_error_found
        jmp      done

done:
        mov      al, 0          ; return code of 0
        mov      ah, 4ch       ; function code for exit to os
        int      21h
prog    endp
        end      prog
```

File Processing Homework Program Output

```
C:\WINDOWS\system32\cmd.exe

C:\Users\Jade\Desktop\ICS312~1\homework\hw14>dir
Volume in drive C has no label.
Volume Serial Number is 6C86-A0A6

Directory of C:\Users\Jade\Desktop\ICS312~1\homework\hw14

10/07/2009  01:12 AM    <DIR>          .
10/07/2009  01:12 AM    <DIR>          ..
10/07/2009  01:12 AM             3,925 hw14.asm
10/07/2009  01:07 AM            58,123 hw14.docx
10/07/2009  01:07 AM           85,940 Jade's ICS 312 hw14.pdf
02/28/2001  01:11 AM             8,227 Pcmac.inc
03/05/2001  10:43 PM           11,659 UTIL.DOC
02/28/2001  12:55 AM             8,863 UTIL.LIB
             6 File(s)          176,737 bytes
             2 Dir(s)  24,855,203,840 bytes free

C:\Users\Jade\Desktop\ICS312~1\homework\hw14>ml hw14.asm util.lib
Microsoft (R) Macro Assembler Version 6.14.8444
Copyright (C) Microsoft Corp 1981-1997. All rights reserved.

Assembling: hw14.asm

Microsoft (R) Segmented Executable Linker Version 5.60.339 Dec  5 1994
Copyright (C) Microsoft Corp 1984-1993. All rights reserved.

Object Modules [.obj]: hw14.obj
Run File [hw14.exe]: "hw14.exe"
List File [nul.map]: NUL
Libraries [.lib]: "util.lib"
Definitions File [nul.def]:

C:\Users\Jade\Desktop\ICS312~1\homework\hw14>hw14.exe
Humpty Dumpty sat on the wall.
Humpty Dumpty had a great fall.

C:\Users\Jade\Desktop\ICS312~1\homework\hw14>dir
Volume in drive C has no label.
Volume Serial Number is 6C86-A0A6

Directory of C:\Users\Jade\Desktop\ICS312~1\homework\hw14

10/07/2009  01:13 AM    <DIR>          .
10/07/2009  01:13 AM    <DIR>          ..
10/07/2009  01:12 AM             3,925 hw14.asm
10/07/2009  01:07 AM            58,123 hw14.docx
10/07/2009  01:13 AM             1,476 hw14.exe
10/07/2009  01:13 AM             580 hw14.obj
10/07/2009  01:07 AM           85,940 Jade's ICS 312 hw14.pdf
10/07/2009  01:13 AM             63 MYFILE.TXT
02/28/2001  01:11 AM             8,227 Pcmac.inc
03/05/2001  10:43 PM           11,659 UTIL.DOC
02/28/2001  12:55 AM             8,863 UTIL.LIB
             9 File(s)          178,856 bytes
             2 Dir(s)  24,855,199,744 bytes free

C:\Users\Jade\Desktop\ICS312~1\homework\hw14>more MYFILE.TXT
Humpty Dumpty sat on the wall.
Humpty Dumpty had a great fall.

C:\Users\Jade\Desktop\ICS312~1\homework\hw14>
```